
StructDualDynProg.jl Documentation

This packages aims at providing an implementation of SDDP that is both efficient and modular/flexible.
It features the following:


	Support for unfeasible problem by generating a feasibility cut.

	Support for unbounded problem by using an unbounded ray.

	Support for a variety of cut pruning algorithm through the CutPruners [https://github.com/JuliaPolyhedra/CutPruners.jl] package.

	Support for any linear or conic solvers available through MathProgBase [https://github.com/JuliaOpt/MathProgBase.jl]; see JuliaOpt’s webpage [http://www.juliaopt.org/] for a list.

	Support modeling the problem using the StructJuMP modeling interface.

	Support specifying the problem using a low-level interface. This is used for example by the EntropicCone [https://github.com/blegat/EntropicCone.jl] package.



The SDDP algorithm can be run from any node of the lattice of problems using the following function:

SDDP(g::AbstractSDDPTree, num_stages; K::Int, stopcrit::AbstractStoppingCriterion, verbose, pathsampler, ztol, stopatinf, mergepaths, forwardcuts, backwardcuts)





This lattice can be built from a StructJuMP model using the following function:

model2lattice(m::JuMP.Model, num_stages, solver, pruningalgo::CutPruners.AbstractCutPruningAlgo, cutgen::AbstractOptimalityCutGenerator, detectlb::Bool, newcut::Symbol)

Stopping Criterion

stop(s::AbstractStoppingCriterion, stats::AbstractSDDPStats, totalstats::AbstractSDDPStats)
OrStoppingCriterion
AndStoppingCriterion
IterLimit
CutLimit
Pereira









          

      

      

    

  

    
      
          
            
  
A first example : Production Planning

In this quick start guide, we show how to run the FAST quick start example [https://web.stanford.edu/~lcambier/fast/demo.php] using this package.
We guide you through each step of the modeling separately.
The full example can be found in the test [https://github.com/blegat/StructDualDynProg.jl/blob/master/test/optimize_stock.jl].

We start by setting the different constants

const num_stages = 2
const numScen = 2
const C = 1
const P = 2
const d = [2, 3]





We now model the master problem using StructJuMP [https://github.com/StructJuMP/StructJuMP.jl].

using StructJuMP
m1 = StructuredModel(num_scenarios=numScen)
@variable(m1, x >= 0)
@objective(m1, Min, C * x)





For each of the two scenarios we need to create a StructJuMP [https://github.com/StructJuMP/StructJuMP.jl] model specifying that m1 is the parent and that the scenario has probability 1/2.

for ξ in 1:numScen
    m2 = StructuredModel(parent=m1, prob=1/2, id=ξ)
    @variable(m2, s >= 0)
    @constraints m2 begin
        s <= d[ξ]
        s <= x
    end
    @objective(m2, Max, P * s)
end





This structured model need to be transformed into an appropriate structure to run SDDP on it.
This is achieved by model2lattice:

using GLPKMathProgInterface
const solver = GLPKMathProgInterface.GLPKSolverLP()
using CutPruners
const pruner = AvgCutPruningAlgo(-1)
using StructDualDynProg
lattice = model2lattice(m1, num_stages, solver, pruner)





In this example, we have chosen the GLPK [https://github.com/JuliaOpt/GLPKMathProgInterface.jl/] solver but you can use any LP solver listed in the table of the JuliaOpt’s webpage [http://www.juliaopt.org/].

You can now run the sddp algorithm on it using SDDP:

sol = SDDP(lattice, num_stages, K = 2, stopcrit = Pereira(0.1) | IterLimit(10))





We are using 2 forward paths per iteration and we stop either after 10 iterations or once the pereira criterion is satisfied with $\alpha = 0.1$.

We can verify that the algorithm have found the right value by inspecting the solution:

@show sol.objval # sol.objval = -2.0









          

      

      

    

  

    
      
          
            
  
Hydro Thermal Scheduling

In this tutorial, we show how to run the FAST tutorial example [https://web.stanford.edu/~lcambier/fast/tuto.php] using this package.
The big difference between this example and the example quickstart example is that in this example we will model serial independence.
There will be 5 stages and 2 scenarios per stages except for the first stage which has only one scenario.
Each pair of scenario will have the same parent.

We start by setting the constants:

const num_stages = 5
const numScen = 2
const C = 5
const V = 8
const d = 6
const r = [2, 10]





We now create a matrix to store all the variables of all the models.
This allows us to use the variables of other models from a given model.
We also create an array of the first model of each stage to give play the role of parent for the models of the next stage.

using StructJuMP
x = Matrix{JuMP.Variable}(num_stages, numScen)
y = Matrix{JuMP.Variable}(num_stages, numScen)
p = Matrix{JuMP.Variable}(num_stages, numScen)
models = Vector{JuMP.Model}(num_stages)





Now, we create all the models.
Note that each model declares that its parent is the first model (i.e. the model ξ == 1) of the previous stage.
Hence if it is not the first model, it also declares that it has the same children than the first model of its stage.
This is how serial independence is modeled in StructJuMP [https://github.com/StructJuMP/StructJuMP.jl].

for s in 1:num_stages
    for ξ in 1:(s == 1 ? 1 : numScen) # for the first stage there is only 1 scenario
        if s == 1
            model = StructuredModel(num_scenarios=numScen)
        else
            model = StructuredModel(parent=models[s-1], prob=1/2, same_children_as=(ξ == 1 ? nothing : models[s]), id=ξ, num_scenarios=(s == num_stages ? 0 : numScen))
        end
        x[s, ξ] = @variable(model, lowerbound=0, upperbound=V)
        y[s, ξ] = @variable(model, lowerbound=0)
        p[s, ξ] = @variable(model, lowerbound=0)
        if s > 1
            @constraint(model, x[s, ξ] <= x[s-1, 1] + r[ξ] - y[s, ξ])
        else
            @constraint(model, x[s, ξ] <= mean(r) - y[s, ξ])
        end
        @constraint(model, p[s, ξ] + y[s, ξ] >= d)
        @objective(model, Min, C * p[s, ξ])
        # models[s] contains the first model only
        if ξ == 1
            models[s] = model
        end
    end
end





We now create the lattice, note that the master problem is models[1].

using GLPKMathProgInterface
const solver = GLPKMathProgInterface.GLPKSolverLP()
using CutPruners
const pruner = AvgCutPruningAlgo(-1)
using StructDualDynProg
lattice = model2lattice(models[1], num_stages, solver, pruner)





The SDDP algorithm can now be run on the lattice:

sol = SDDP(lattice, num_stages, K = 16, stopcrit = Pereira(2., 0.5) | IterLimit(10))

